EulerAngles

In[1221]:=
```
Clear["Global`*"]
```

Part 1: Euler Angle rotation

In[1222]:=
```
Rzphi[t] = {
    {Cos[phi[t]], -Sin[phi[t]], 0},
    {Sin[phi[t]], Cos[phi[t]], 0},
    {0, 0, 1}};
Rytheta[t] = {
    {Cos[theta[t]], 0, Sin[theta[t]]},
    {0, 1, 0},
    {-Sin[theta[t]], 0, Cos[theta[t]]}};
Rzpsi[t] = Rzphi[t] /. phi[t] → psi[t];
```

In[1225]:=
```
MatrixForm[Rzphi[t]]
MatrixForm[Rytheta[t]]
MatrixForm[Rzpsi[t]]
```

Out[1225]//MatrixForm=

$$\begin{pmatrix} \text{Cos[phi[t]]} & -\text{Sin[phi[t]]} & 0 \\ \text{Sin[phi[t]]} & \text{Cos[phi[t]]} & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Out[1226]//MatrixForm=

$$\begin{pmatrix} \text{Cos[theta[t]]} & 0 & \text{Sin[theta[t]]} \\ 0 & 1 & 0 \\ -\text{Sin[theta[t]]} & 0 & \text{Cos[theta[t]]} \end{pmatrix}$$

Out[1227]//MatrixForm=

$$\begin{pmatrix} \text{Cos[psi[t]]} & -\text{Sin[psi[t]]} & 0 \\ \text{Sin[psi[t]]} & \text{Cos[psi[t]]} & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

In[1228]:=
```
amat = Rzphi[t].Rytheta[t].Rzpsi[t];
```

In[1229]:=
```
MatrixForm[Simplify[amat.{0, 0, 1}]]
```

Out[1229]//MatrixForm=

$$\begin{pmatrix} \text{Cos[phi[t]] Sin[theta[t]]} \\ \text{Sin[phi[t]] Sin[theta[t]]} \\ \text{Cos[theta[t]]} \end{pmatrix}$$

In[1230]:=
```
rbody = {x, y, z};
```

In[1231]:=
```
rspace = Simplify[amat.rbody];
```

In[1232]:=
```
vspace = FullSimplify[D[rspace, t]];
```

In[1233]:=

```
vbody = Simplify[Transpose[amat].vspace]
```

Out[1233]=
```
{(-y Cos[theta[t]] + z Sin[psi[t]] Sin[theta[t]]) phi'[t] -
   y psi'[t] + z Cos[psi[t]] theta'[t],
  (x Cos[theta[t]] + z Cos[psi[t]] Sin[theta[t]]) phi'[t] + x psi'[t] -
   z Sin[psi[t]] theta'[t], -(y Cos[psi[t]] + x Sin[psi[t]]) Sin[theta[t]] phi'[t] +
   (-x Cos[psi[t]] + y Sin[psi[t]]) theta'[t]}
```

In[1234]:=
```
Lbodydensity = Simplify[Cross[rbody, vbody]]
```

Out[1234]=
```
{-(x z Cos[theta[t]] + ((y^2 + z^2) Cos[psi[t]] + x y Sin[psi[t]]) Sin[theta[t]]) phi'[t] -
   x z psi'[t] + (-x y Cos[psi[t]] + (y^2 + z^2) Sin[psi[t]]) theta'[t],
  (-y z Cos[theta[t]] + (x y Cos[psi[t]] + (x^2 + z^2) Sin[psi[t]]) Sin[theta[t]]) phi'[t] -
   y z psi'[t] + ((x^2 + z^2) Cos[psi[t]] - x y Sin[psi[t]]) theta'[t],
  ((x^2 + y^2) Cos[theta[t]] + z (x Cos[psi[t]] - y Sin[psi[t]]) Sin[theta[t]]) phi'[t] +
   (x^2 + y^2) psi'[t] - z (y Cos[psi[t]] + x Sin[psi[t]]) theta'[t]}
```

In[1235]:=
```
Lbody = Simplify[Expand[Lbodydensity] /. {x * y → 0, x * z → 0, y * z → 0,
      x^2 → (i2 + i3 - i1) / 2, y^2 → (i3 + i1 - i2) / 2, z^2 → (i1 + i2 - i3) / 2}];
MatrixForm[
 Lbody]
```

Out[1236]//MatrixForm=
```
( -i1 Cos[psi[t]] Sin[theta[t]] phi'[t] + i1 Sin[psi[t]] theta'[t] )
(  i2 (Sin[psi[t]] Sin[theta[t]] phi'[t] + Cos[psi[t]] theta'[t])  )
(              i3 (Cos[theta[t]] phi'[t] + psi'[t])                )
```

In[1237]:=
```
Ibody = DiagonalMatrix[{i1, i2, i3}];
MatrixForm[Ibody]
```

Out[1238]//MatrixForm=
```
( i1  0   0  )
( 0   i2  0  )
( 0   0   i3 )
```

In[1239]:=
```
omegabody = Simplify[Inverse[Ibody].Lbody];
MatrixForm[omegabody]
```

Out[1240]//MatrixForm=
```
( -Cos[psi[t]] Sin[theta[t]] phi'[t] + Sin[psi[t]] theta'[t] )
(  Sin[psi[t]] Sin[theta[t]] phi'[t] + Cos[psi[t]] theta'[t] )
(             Cos[theta[t]] phi'[t] + psi'[t]                )
```

Part 2: Euler Eqs. for free rigid body (L = T - V with V=0)

In[1241]:=
```
T = Simplify[(1 / 2) * omegabody.Ibody.omegabody]
```

Out[1241]=
```
1/2 (i3 (Cos[theta[t]] phi'[t] + psi'[t])^2 +
   i2 (Sin[psi[t]] Sin[theta[t]] phi'[t] + Cos[psi[t]] theta'[t])^2 +
   i1 (Cos[psi[t]] Sin[theta[t]] phi'[t] - Sin[psi[t]] theta'[t])^2)
```

```
In[1242]:= ppsi = D[T, psi'[t]]

Out[1242]= i3 (Cos[theta[t]] phi'[t] + psi'[t])

In[1243]:= Simplify[ppsi - i3 * omegabody[[3]]]

Out[1243]= 0

In[1244]:= fpsi = Simplify[D[T, psi[t]]]

Out[1244]= -(i1 - i2) (Sin[psi[t]] Sin[theta[t]] phi'[t] + Cos[psi[t]] theta'[t])
             (Cos[psi[t]] Sin[theta[t]] phi'[t] - Sin[psi[t]] theta'[t])

In[1245]:= Simplify[% - (i1 - i2) * omegabody[[1]] * omegabody[[2]]]

Out[1245]= 0

In[1246]:= (* Hence i3*D[omegabody[[3]],t] == (i1-i2)*omegabody[[1]]*omegabody[[2]] *)

In[1247]:= (* In place of the other two Lagrange Eqs. of motion,
           can just use cyclic permutations of this. *)

In[1248]:= Part 3 : Free symmetric rigid body (L = T - V with V = 0 and i1 = i2)

In[1248]:= Tsym = T /. i2 → i1

Out[1248]= 1/2 (i3 (Cos[theta[t]] phi'[t] + psi'[t])^2 +
              i1 (Sin[psi[t]] Sin[theta[t]] phi'[t] + Cos[psi[t]] theta'[t])^2 +
              i1 (Cos[psi[t]] Sin[theta[t]] phi'[t] - Sin[psi[t]] theta'[t])^2)

In[1249]:= fphi = Simplify[D[Tsym, phi[t]]]

Out[1249]= 0

In[1250]:= fpsi = Simplify[D[Tsym, psi[t]]]

Out[1250]= 0

In[1251]:= Pphi = FullSimplify[D[Tsym, phi'[t]]]

Out[1251]= 1/2 ((i1 + i3 + (-i1 + i3) Cos[2 theta[t]]) phi'[t] + 2 i3 Cos[theta[t]] psi'[t])

In[1252]:= Ppsi = FullSimplify[D[Tsym, psi'[t]]]

Out[1252]= i3 (Cos[theta[t]] phi'[t] + psi'[t])

In[1253]:= Lspace = {0, 0, Lmag}

Out[1253]= {0, 0, Lmag}

In[1254]:= Lbody = FullSimplify[Transpose[amat].Lspace]

           {-Lmag Cos[psi[t]] Sin[theta[t]], Lmag Sin[psi[t]] Sin[theta[t]], Lmag Cos[theta[t]]}
```